
1.0 Executive Summary
The radial preload requirement for the original toroidal field coil (2 toroidal turns x 6 radial turns) has been fully documented in an earlier memo
. However, recently the project has elected to pursue a narrower configuration (3 toroidal turns x 4 radial turns with a smaller conductor) used in conjunction with cast SS wedges in order to eliminate machining the WP inboard leg. 

A 3D coupled electromagnetic-structural ANSYS
 model of this latest NCSX TF and Modular Coil (MC) systems is assembled and used to determine Lorentz forces, deformations and, to some extent stresses within the TF magnet from one particularly limiting equilibrium current set. The analysis focuses on the Modular Coil Fields at S3 state, when TF currents are low and MC currents are high but negative in some or all coils. Field analyses using current-sticks
 has shown that some of the TF coils have a positive net radial force. This analysis is designed to check those earlier results [3] and draw a more detailed picture of the deflections and radial preload requirement.
A simulation which includes frictional contact between wedged faces of just two of the six coils indicates that with a 4000 lb/bracket preload, the threshold friction coefficient is 0.24. A second analysis with updated TF coil pack smeared orthotropic properties and more realistic boundary conditions (BCs) indicate a threshold friction coefficient of 0.10. This dramatic drop in the minimum friction coefficient requirement is consistent with the revised boundary conditions and suggests plenty of margin in the 4000 lb/bracket preload. Even halving the preload to 2000 lb only increases the minimum surface friction coefficient to 0.19, which should be easy to achieve.
If relying on friction alone (as a “belt”) causes some trepidation, then adding a mechanical fastener capable of transmitting ~1000 lb of shear at the extremes of the wedged interfaces, would proved some margin (and the “suspenders”) to the structure.
2.0 Assumptions and Notable Concerns
2.1 The stability of the TF coil system under light TF operating currents will rely on a mechanical preload and friction at the wedged surfaces to hold back the TF coils with a positive radial Lorentz force. An assumed friction coefficient of 0.3 is used in the analysis, which should be relatively easy to achieve. Most materials have friction coefficients greater than 0.3, and achieving a low friction surface usually takes some special effort. The subject coil currents indicate that the minimum friction coefficient requirement to a mere 0.1 when a 4000 lb/bracket preload is applied. Halving the radial preload results in a doubling of the minimum friction coefficient to 0.19. 
2.2 The coil currents used here are chosen because they produce positive radial forces on some TF coils, tending to eject them from the coil system. It is assumed that A. Brooks’ design-space search has identified the worst-case condition for producing slippage between coils. 

2.3 The computational demands of modeling six contact surfaces for this 120˚ field period model are too great for the resources available to Myatt Consulting, Inc. However, the EM analysis of [1] shows that a single contact surface in the middle of the six-coil model is the most likely place for slippage based on poloidal torque summations. 

3.0 Analysis

The analysis is based on the 3D model shown in Fig. 3.0-1. There are six TF coils and six MCs which make up one 120˚ “field period.” Smeared TF coil winding packs (WP) are bonded to cast SS wedged. General surface to surface contact elements allow for contact, separation and sliding at the θ=0 symmetry plane. All other wedges are bonded to their neighbor. 
The electromagnetic (EM) model is loaded by applying the following currents: NI(TF)=-18.43 kA-t, NI(MC1)=820kA-t, NI(MC2)=830 kA-t and  NI(MC3)=730 kA-t. The structural model is loaded by importing the Lorentz forces calculated by the EM model and the applied radial preload (located at ±52” about the equatorial plane). The model is very similar to the one described in [1] with the changes to the TF coils and support wedges described in [4]. Salient drawings are included here as Attachment 6.1.
Deformations and stresses are postprocessed in order to assess the stability of the wedged surfaces (i.e., conditions which produce stick-slip behavior). 

Fig. 3.0-1 Isometric View, One Field Period of TF & Modular Coils

4.0 Results

The results of the new TF WP with SS wedges and a 4000 lb per bracket preload are presented below. 
Fig. 4.0-1 is a plot of the deformed coil & wedges, which shows local contact at the upper and lower extremes and a 5 mil gap at the equator.
Fig. 4.0-2 is a plot of the relative motion at the θ=0, TF3/4 contact surface. The contour values represent the amount of tangential relative motion and indicate a maximum of 39 μm or 1.5 mils. This maximum occurs in the red regions mid-way between the Z=0 equatorial plane and the two extremes of the interface. It is worth noting that this is about half of the relative motion found in the previous analysis of the larger WP without SS wedges [1]. Sliding at the extremes of the interface is nil, as represented by the “MN” symbol which correspond to a slip of 10-14 m.

The ANSYS plot title also lists some interesting results about the status of the contact elements: 2.1% are stuck, 1.7% are sliding, and 96% are open. In addition, of the contact elements which are closed (i.e., in contact and sliding or stuck) the average ratio of friction force to normal force is 0.24. This result is based on a very conservative calculation which goes as follows:

1. Determine the resultant frictional force for each contact element, F_S = (F_Y2 + F_Z2)0.5.

2. Divide F_S by the normal force (F_N) for each contact element, MUJ = F_S/F_N.
3. Determine the average friction coefficient, Σ(MUJ)/(No. of closed contact elements).

Another approach is to uses global force summations from the ETABLE sum shown in Fig. 4.0-3, which goes as follows:

1. Calculate the resultant frictional force Fs (sliding or stuck) from all contact elements above the equator.
· Fs = (Σ(F_Y)2+ Σ(F_Z)2)0.5 = (13602 + 39402)0.5 = 4170 N
2. Calculate the total contact force from all contact elements above the equator.
· Fn = 55540 N

3. Divide Fs by Fn to determine the average friction coefficient required to maintain equilibrium.

· μmin = (4170/55540) = 0.075

This global force ratio of 0.075 is substantially smaller than the local average fiction coefficient value of 0.24 because some of the Fy and Fz forces have opposite signs which reduce the net frictional force. One could think of it this way: Eliminate macro-level slip by assuring a wedge surface friction coefficient of 0.24 or greater. Eliminate global-level slip by assuring a wedge surface friction coefficient of 0.08 or greater. It is worth noting that the assumed friction coefficient used in the analysis (0.3) is from conservative engineering judgment based on a survey of numerous materials with sliding contact (Marks’ Handbook). Increasing the preload will reduced the minimum required friction coefficient if 0.24 is a concern. Also, any mechanical fasteners at the extremes of the wedged interfaces would supplement or eliminate this frictional component. This mechanical device must restrain 4170 N (~1000 lb) in shear to carry the differential loads developed at this limiting operating condition.

Fig. 4.0-1 Toroidal Displacement Contours [m] Superimposed on Exaggerated Deformed Shape
Fig. 4.0-2 Relative Motion at TF3/4 Interface

Fig. 4.0-3 Summation of Various ETABLE Components from Upper Half Contact Elements

Fig. 4.0-4 is included to show the stress intensity in the smeared WP as a result of this particular coil current set plus the 4000 lb/bracket mechanical preload operating condition. The plot shows that the highest stresses appear at the lateral displacement boundary conditions, which make the 66 MPa value suspect. Away from these lateral constraints, the plot indicates larger regions at a stress level of 25 MPa (in the transition regions of all TF coils). Even with the understanding that these are smeared stresses which must be “marked-up” to determine the stress in the Cu conductor, the [image: image6.png]N

magnitude is low enough to not be a concern (Local Primary Membrane limit: 270 MPa).

Fig. 4.0-4 Stress Intensity in the Smeared TF WP

TFCs & MCs from 0.048T S3 Field State, 4000 lb/bracket Preload (μ=0.3 Friction at TF3/4 surface)
4.1 Preload Analysis with Revised Smeared Properties

The results presented in Sect. 4.0 are based on out-dated smeared winding pack properties. A recent change in the TF conductor insulation system, from fully-bonded to a Kapton-glass slip-plane, has triggered a revision to the smeared properties used in all global analyses
. Since it is not clear what effect, if any, this change will have on the radial preload requirements, the six-coil model is revised and rerun. 

It should be noted that the earlier version of the model (such as reported in 4.0) uses isotropic properties for the smeared WP. In this revised model, the orthotropic properties from [4] are applied. This requires defining a series of local coordinates (~10 per coil) for each WP segment (straights in the inboard leg and arcs elsewhere), which requires a fairly tedious effort. Conversely, all of these local systems can be approximated by a single toroidal system. Of course, the TF coils are not precisely toroidal (their minor radius is not a single value). However, the simplification is so trivial and the results are so close to “right” that the approach is applied here (see Fig. 4.1-1).

Fig. 4.1-1 Orthotropic WP Properties Parallel to Toroidal System Illustrated by Element CS Triads

Some difficulties in convergence have lead to another change: instead of supporting the –Y coils from the top and +Y coils from the bottom (as shown in Fig. 3.0-1), all coils are supported from the bottom. (It is very likely that this was the source of the convergence problem. It also produces some asymmetries that are non-physical and would not be present in the actual coil.) 

[image: image8.jpg]FoLD

FoLD

3 s 5 | 8 10 I S 12 | 13 | 14 I 15
NO. REVISION BY CH SUP APPROVED DATE
fe——— | ] 3/4 ——
RI
fe—— 4 3/|6 ——————
2 3/4 ——=—]
]
)
o (2 3/4)
10.00 \
/
* \ [ Y
AN 3/8 — o
R?26 15/16 /\ 733 3/8
8 R 1/8 ——f=—
46 .9
I I ; ; ; ; ; X
] I SECTION A-A
SCALE 1.000
21
28 13/16
54
Y 1 i Y
[03 3/4
571 9/16
Y
Y
R31 5/8
b |
OT APPROVED
)\ 2
| | TF_CASTING_STUDY | TF COIL NOSE WEDGE CASTING RIGHT | STN STL
PART NOMENCLATURE oTY
NO. DRAWING NO OR DESCRIPTION MATERIAL |oroD
i PARTS LIST
COMPUTER GENERATEDI contmaL Fives: PRINCETON PLASMA PHYSICS LABORATORY
MANUAL CHANGES PRINCETON UNIVERSITY
NoT PeRMITTED |UNLEss otherwisE speciFiEn]  NATIONAL COMPACT STELLARATOR EXPERIMENT
T W (SR GO,
153.3 Ibs P07 SEating DRming O | BREAX suaRP EDGES .005/.020 TF NOSE CASTING DETAIL
MODEL NAME SCALE 0.250 | ToLERANCES NON-CUMULATIVE | DSN:J. RUSHINSKI DRAWING NO:
NEXT ASSEMBLY DECIMAL- INCH| FRACTIONS | CHK:
RELEASE LEVEL: TF-CASTING. STUDY S A e TF_NOSE.CASTING
DWG VERSION NO: ENGINEER Lo 13 | ovin r20mer-ira] SUPY: SHEET | OF | REV
3 43 5 | 8 | 10 Il z 12 I3 | | 4 NCSX_PART _FORMAT _E

foLb

1]

< NCSX-TF_NOSE_CASTING




Fig. 4.1-2 shows a plot with these vertical displacement constraints highlighted, as well as the radial preload vectors, EM vectors, and lateral support points. The embedded ANSYS plot title also lists the coil currents associated with this EM loading.
Fig. 4.1-2 Force and Displacement Boundary Conditions

Fig. 4.1-3 is a plot of the radial deformations superimposed on a greatly exaggerated deformed shape plot from the radial preload condition. Ignore the embedded plot title which indicates that coil currents are present (“STEP=1” contains preload forces only). The plot shows how the coil sets contact each other at the extremes of the interface while producing a small gap in between. The back leg moves in by ~1.4 mm.

Fig. 4.1-3 Gaps, Contact and Radial Deformations (Smeared Properties of Slip-Plane Insulation)
(4000 lb/bracket Preload Load Case)
Fig. 4.1-4 is a similar plot of the radial deformations superimposed on a greatly exaggerated deformed shape plot. But in this case, the reference EM forces (coil amp-turns embedded in plot title) are included along with the radial preload (STEP=2). A close-up of the contact surface shows perfect radial displacement continuity (no color jump which signifies a stuck condition). 

Fig. 4.1-4 Gaps, Contact and Radial Deformations (Smeared Properties of Slip-Plane Insulation)

TFCs & MCs from 0.048T S3 Field State, 4000 lb/bracket Preload (μ=0.3 Friction at TF3/4 surface)
The condition of the contact surface is quantified by a survey of the gap element “status.” Fig. 4.1-5 is a plot of the relative motion across each gap element. Dots signify no relative motion, while lines are vector-like and indicate the magnitude and direction of the relative motion. Notice that the relative motion at the extremes is represented by blue dots (relative displacements of order 10-8 m). Somewhat above the equatorial plane, the relative motion is a maximum at 0.06 mm. The embedded plot title lists other salient results:

· 5.9% of the gaps are “Stuck” (closed and not sliding)

· 0.5% of the gaps are “Sliding” (closed and sliding)

· 93.7% of the gaps are “Open”

· The ratio of shear force to normal force gives an effective friction coefficient, Mu(ave)=0.1
It is interesting to note that the average friction coefficient has dropped from 0.24 (Fig. 4.0-2) to 0.1, and is consistent with the 0.075 hand-calculation shown in sect. 4.0. The best physical explanation must be tied to the vertical constraint difference between the two models. The previous analysis (4.0) has boundary conditions which produce relative shear forces even when only in-plane radial preload forces are applied. In this later analysis, the BCs cannot induce such frictional shear forces. Consequently, the frictional requirement goes down and the 4000 lb/bracket may be higher than it needs to be. 

A couple additional analyses are made in order to gain a better understanding of the relationship between radial preload (FPL) and average surface friction coefficient (μave). Fig. 4.1-6 summarizes this result with a simple line plot. As the preload is reduced from 4000 lbf/bracket, the average friction coefficient required to maintain a “stuck” wedge contact interface increases. Notice that when the preload is reduced to 2000 lbf then the friction coefficient increases to 0.19.
Fig. 4.1-5 Relative Motion at TF3/4 Interface (Smeared Properties of Slip-Plane Insulation)

TFCs & MCs from 0.048T S3 Field State, 4000 lb/bracket Preload (μ=0.3 Friction at TF3/4 surface)

This plot demonstrates that the 4000 lb/bracket radial preload is sufficient to maintain a stuck interface (albeit limited to small patches at the extremes) even though there is a net radial force on one (or more) of the TF coils. 

5.0 Summary & Commentary
At certain times during the current pulse, the net EM load on some of the TF coils is a positive radial force. Therefore, the project is obliged to provide some inward radial preload in order to ensure a structurally stable TF coil system. 

The analysis shows that when the radial preload is 4000 lb/bracket (two per TF coil), then a friction coefficient of >0.10 at the wedge-to-wedge contact surface will maintain a no-slip condition. Analyses at lower preload levels suggest that a 2000 lb/bracket preload would also work, and results in a modest increase in the minimum friction coefficient to 0.19. In general engineering terms, these are both relatively low values, and should be easy to achieve with possibly no design effort. As added insurance against progressive relative motion, a mechanical device could be designed to transmit ~1000 lb of shear force at the extremes of the wedged interfaces.
6.0 Attachments

6.1 Drawings
[image: image13.png]N
Fie Hop

1 ANSYS 8.1

NOV 3 2004

14:15:51

NODAL SOLUTION

STEP=2

Hard Contact SUB =12

(Top & Bottom) TIME=2

uy

RSYS=1

DMX =.001166

SMN =-.674E-04

SMX =.673E-04
-.674E-04

- -.525E-04
-.375E-04
-.225E-04
-.754E-05

Toroidal Gap
=
[
[
o .743E-05
=
[

(Equator)

.224E-04
.374E-04
.523E-04
.673E-04

tfmods30







6.2 ANSYS Input File

/batch

rn=5

!/filnam,tfmodb3%rn%

!/show,tfmodb3%rn%,grp

resume,tfmodb3%rn%,db

*if,1,eq,1,:2000

/prep7

/com

/com Electromagnetic analysis of the NCSX TF and Modular Coil System

/com Structural analysis of the TF Coil System (No OOP structure...just inboard wedging and BCs)

/com Derived from tfmod2.dat, Coils made as smeared without ground wrap or sidewall insulation.

/com This coarser model designed to run with wedging contact elements to determine potential for stick-slip

/com Geometry defined by new 3x4 conductor pack and wedges

/com Solve the Conduction and field problem, then solve the structural problem

/com 

/com

/com Model by Leonard Myatt, Myatt Consulting, Inc.

/com   leonard.myatt@myattconsulting.com, 508-520-4590

/com

/com

/com Run History

/com

/com 30: TF Only...looks OK

/com 31: EM Only: TF & MC currents from Art Brooks (TF:+25.4 kA-t, M1/M2/M3:-850/+850/-850 kA-t)

/com 32: EM Only: TF & MC currents from Art Brooks (TF:-18.43 kA-t, M1/M2/M3:+820/+830/+730 kA-t)

/com Revise Smeared Properties & 3/8" GW thickness

/com 33: TF & MC currents from Art Brooks (TF:-18.43 kA-t, M1/M2/M3:+820/+830/+730 kA-t), k_dens=3

/com 34: TF & MC currents from Art Brooks (TF:-18.43 kA-t, M1/M2/M3:+820/+830/+730 kA-t), k_dens=2

/com 35: TF & MC currents from Art Brooks (TF:-18.43 kA-t, M1/M2/M3:+820/+830/+730 kA-t), k_dens=1, UZ=0 @ Z=0

/com 35: TF & MC currents from Art Brooks (TF:-18.43 kA-t, M1/M2/M3:+820/+830/+730 kA-t), nix toln & ftol on real

/com 35: TF & MC currents from Art Brooks (TF:-18.43 kA-t, M1/M2/M3:+820/+830/+730 kA-t), Key2=1 Lagrange Mult.

/com 35: TF & MC currents from Art Brooks (TF:-18.43 kA-t, M1/M2/M3:+820/+830/+730 kA-t), Key2=2, k_dens=1

/com

/com

/com Contact and Mesh Density Parameters

k_nl=1                   ! 0: Wedged surfaces held to UY=0, 1: Generate Flex-Rigid Contact Interface

k_dens=1                 ! number of elements across WP (3 is good, smaller can cause meshing problems)

mu_sw=0.3                ! sidewall-to-sidewall friction coefficient

/com Misc Parameters

k=0.0254                 ! english to si conversion factor

pi=acos(-1)              ! pi

mu0=4*pi*1e-7            ! Mu0

*afun,deg                ! use degrees in trig functions

t=0.0001                 ! a tiny length

th=0.1                   ! a tiny angle

/com Nominal Performance Parameters

f_preload=4000/0.2248    ! applied radial preload per TF Bracket x 2 brackets per coil

z_preload=52*k           ! vertical location of applied preload

b0=0.5                   ! nominal flux density (one way of specifying TF current)

r0=1.4                   ! major radius

ntf=18                   ! number of tf coils

nsym=3                   ! toroidal symmetry

arc=360/ntf              ! angular extent of 1 tf coil

emsym,nsym               ! electromagnetic symmetry about Z

dtmp=0!-215                ! differential temperature (WRT room temp)

/com

/com Coil Currents

/com

n_lay=3                  ! number of radial layers

n_pan=4                  ! number of pancakes

ntf=18                   ! number of tf coils

i_tf=-18.43e3 !5e6*b0*r0/ntf       ! total current in each smeared tf winding pack, A-turns/coil

i_mc1=820e3/20           ! conductor current in MC1, Amps

i_mc2=830e3/20           ! conductor current in MC2, Amps

i_mc3=730e3/18           ! conductor current in MC3, Amps

/com

/com element types

/com

et,1,98,2             ! Wedges (UX, UY, UZ)

et,2, 5,1             ! Modular Coil WP (TEMP, VOLT, MAG)

et,3,98,1             ! TF WP (TEMP, VOLT, MAG)

/com

/com graphics keys

/com

/pnum,mat,1

/num,1

/dist

/focus

/vup,1,z

/view,1,,-1

/com

/com tf conductor, ignoring radii (in local CS vernacular)

/com

dr_con=0.966*k                        ! Conductor build in thickness (0.434*k)

dz_con=0.709*k                        ! Conductor build in height (1.50*k)

dri_con=0.312*k                       ! inside thickness of cooling channel

dzi_con=0.312*k                       ! inside height of cooling channel

a_con=dr_con*dz_con-dri_con*dzi_con   ! conductor metal area

/com

/com Insulation

/com

t_tw=0.049*k              ! turn wrap insulation thk

t_pan=0.0*k               ! pancake insulation thk (was 0.030", can be 0.0)

t_lay=0.0*k               ! layer insulation thk (can be 0.0)

t_gw=0.375*k               ! module over-wrap thickness

t_gap=t_gw/10             ! height of gap in outboard leg for current application

/com

/com WP Build

/com

dr_wp=n_lay*(dr_con+2*t_tw)+(n_lay-1)*t_lay   ! radial build of WP

dz_wp=n_pan*(dz_con+2*t_tw)+(n_pan-1)*t_pan   ! toroidal build of WP

dr_iwp=dr_wp+2*t_gw                           ! radial build of ground-wrapped WP

dz_iwp=dz_wp+2*t_gw                           ! toroidal build of ground-wrapped WP

cel=sqrt(dr_wp**2+dz_wp**2)/k_dens            ! characteristic element size

/com

/com Centers of sweep and inside radii

/com

x11=r0                       ! x position of local 11

z11=0.0                      ! z position of local 11

ri11=r0-(12.494*k+dr_iwp)    ! inside radius of coil in local 11

/com

x12=43.239*k                 ! x position of local 12

z12=28.785*k                 ! z position of local 12

ri12=27.314*k-(t_gw-k/8)                ! inside radius of coil in local 12

/com

x13=53.629*k                 ! x position of local 13

z13=7.141*k                  ! z position of local 13

ri13=51.323*k-(t_gw-k/8)                ! inside radius of coil in local 13

/com

x14=53.629*k                 ! x position of local 14

z14=0.000*k                  ! z position of local 14

ri14=ri13+z13-(t_gw-k/8)                ! inside radius of coil in local 14

/com

x15=58.111*k                 ! x position of local 15

z15=10.850*k                 ! z position of local 15

ri15=47.614*k-(t_gw-k/8)                ! inside radius of coil in local 15

/com

ri16=199.614*k-(t_gw-k/8)               ! inside radius of coil in local 16

x16=(97.051+12.494-203.046)*k! x position of local 16

z16=0.000*k                  ! z position of local 16

/com

x17=58.111*k                 ! x position of local 17

z17=-10.850*k                ! z position of local 17

ri17=ri15                ! inside radius of coil in local 17

/com

x18=58.111*k                 ! x position of local 18

z18=0.000*k                  ! z position of local 18

ri18=ri14                    ! inside radius of coil in local 18

/com

x19=53.629*k                 ! x position of local 19

z19=-7.141*k                 ! z position of local 19

ri19=ri13                ! inside radius of coil in local 19

/com

x20=43.239*k                 ! x position of local 20

z20=-28.785*k                ! z position of local 20

ri20=27.314*k-(t_gw-k/8)                ! inside radius of coil in local 20

/com

/com Swept Angles (and straight-lengths)

/com

dy11=2*z12

dy12=abs(atan((z12-z13)/(x13-x12)))

dy13=90-dy12

dy14=4.482*k                 ! length of section 14

dy15=90-abs(atan((z15-z16)/(x15-x16)))

dy16=2*abs(atan((z15-z16)/(x15-x16)))

dy17=dy15

dy18=dy14

dy19=dy13

dy20=dy12

/com

/com material properties (Toroidal System: X is radial, Y is toroidal and Z is poloidal)

/com

local,103,3,,,,,,,1.4

/com TF Coils

mp, kxx,2,1

mp,murx,2,1

mp,rsvx,2,1

mp,  ex,2,42.3e9

mp,  ey,2,41.2e9

mp,  ez,2,76.6e9

mp,alpx,2,9.5e-6

mp,alpy,2,9.6e-6

mp,alpz,2,13.0e-6

mp,nuxy,2,0.319

mp,nuyz,2,0.284

mp,nuxz,2,0.284

mp, gxy,2,2.1e9

mp, gyz,2,2.1e9   ! required to get beam bending right

mp, gxz,2,2.1e9   ! required to get beam bending right

/com

/com Side Wall insulation (Y is through thickness)

/com

mp,murx,3,1

mp,  ey,3,21e9

mp,  ex,3,28e9

mp,  ez,3,28e9

mp,alpy,3,22e-6

mp,alpx,3,7.9e-6

mp,alpz,3,7.9e-6

mp, gxy,3,6.9e9

mp, gyz,3,6.9e9

mp, gxz,3,9.6e9

mp,nuxy,3,0.21

mp,nuyz,3,0.21

mp,nuxz,3,0.21

mp,dens,3,1800

mp, kxx,3,1

/com

/com Cast SS Wedges

/com

mp,murx,16,1

mp,  ex,16,159e9

mp,alpx,16,13e-6

mp, kxx,16,1

/com

/com coordinate systems

/com

/com Local 11

csys

wpcsys

wpoff,x11,,z11

wprot,,-90

wprot,180

cswpla,11

/com Local 12

csys

wpcsys

wpoff,x12,,z12

wprot,,-90

wprot,180

cswpla,12,1

/com Local 13

csys

wpcsys

wpoff,x13,,z13

wprot,,-90

wprot,180

wprot,dy12

cswpla,13,1

/com Local 14

csys

wpcsys

wpoff,x14,,z14

wprot,,-90

wprot,180

wprot,90

cswpla,14

/com Local 15

dy1215=90

csys

wpcsys

wpoff,x15,,z15

wprot,,-90

wprot,180

wprot,90

cswpla,15,1

/com Local 16

csys

wpcsys

wpoff,x16,,z16

wprot,,-90

wprot,180

wprot,90+dy15

cswpla,16,1

/com Local 17

csys

wpcsys

wpoff,x17,,z17

wprot,,-90

wprot,180

wprot,180+dy16/2

cswpla,17,1

/com Local 18

csys

wpcsys

wpoff,x18,,z18

wprot,,-90

wprot,180

wprot,180+90

cswpla,18

/com Local 19

csys

wpcsys

wpoff,x19,,z19

wprot,,-90

wprot,180

wprot,180+90

cswpla,19,1

/com Local 20

csys

wpcsys

wpoff,x20,,z20

wprot,,-90

wprot,180

wprot,180+90+dy19

cswpla,20,1

/com

/com Smeared Winding Pack

/com

/com Straight Leg

csys,11

wpcsys

block,ri11,ri11+dr_iwp,-dy11/2,,-dz_iwp/2,dz_iwp/2

block,ri11,ri11+dr_iwp,,dy11/2,-dz_iwp/2,dz_iwp/2

/com Local 12,13 Arc

*do,j,12,13

csys,j

wpcsys

ksel,s,loc,x,0.99*ri%j%,1.01*ri%j%

*get,ri%j%,kp,,mnloc,x

vsel,none

cylind,ri%j%,ri%j%+dr_iwp,-dz_iwp/2,dz_iwp/2,,dy%j%

*enddo

/com Top Straight

csys,14

wpcsys

ksel,s,loc,x,0.99*ri14,1.01*ri14

*get,ri14,kp,,mnloc,x

vsel,none

block,ri14,ri14+dr_iwp,,dy14,-dz_iwp/2,dz_iwp/2

/com Local 15,16,17 Arc

*do,j,15,17

csys,j

wpcsys

ksel,s,loc,x,0.99*ri%j%,1.01*ri%j%

*get,ri%j%,kp,,mnloc,x

vsel,none

cylind,ri%j%,ri%j%+dr_iwp,-dz_iwp/2,dz_iwp/2,,dy%j%

*enddo

/com Bot Straight

csys,18

wpcsys

ksel,s,loc,x,0.99*ri18,1.01*ri18

*get,ri18,kp,,mnloc,x

vsel,none

block,ri18,ri18+dr_iwp,,dy18,-dz_iwp/2,dz_iwp/2

/com Local 19,20 Arc

*do,j,20,19,-1

csys,j

wpcsys

ksel,s,loc,x,0.99*ri%j%,1.01*ri%j%

*get,ri%j%,kp,,mnloc,x

vsel,none

cylind,ri%j%,ri%j%+dr_iwp,-dz_iwp/2,dz_iwp/2,,dy%j%

*enddo

/com

/com Glue Up WPs

/com

allsel

vglue,all

cm,wp,volu

/com

/com Add the SS Case

/com

th_case=60                 ! angular extent of wedges in dewedged region

dz_case=103.75*k           ! vertical height of wedges

dy_case=2.75*k             ! max toroidal thickness of wedge

r_case=0!.5*k               ! radius top of case

vsel,none

/com Straight Region

csys,11

wpcsys

block,ri11-0.375*k,ri11+dr_iwp+(7/8)*k,,dy11/2,dz_iwp/2-0.733*k,2*dz_iwp

cm,case,volu

block,ri11,ri11+dr_iwp,,dy11/2,,dz_iwp/2

block,,ri11,,27*k,,dz_iwp/2

vsbv,case,all

cm,case1,volu

/com Arc Region

csys,12

wpcsys

vsel,none

cylind,ri12-0.375*k,ri12+dr_iwp+(7/8)*k,dz_iwp/2-0.733*k,2*dz_iwp,,th_case

cm,case,volu

cylind,ri12,ri12+dr_iwp,,dz_iwp/2,,th_case

vsbv,case,all

cmsel,a,case1

cm,case,volu

csys,11

wpcsys

!block,ri11+dr_iwp+(3/8)*k,12,,(28+13/16)*k,,12

block,ri11+dr_iwp+(3/8)*k,12,,z12,,12

csys

wpcsys

block,,12,,12,dz_case/2,12

block,,12,dz_iwp/2-0.733*k+dy_case,12,,12

wprot,arc/2

block,,12,,12,,12

vsbv,case,all

vglue,all

cm,case,volu

allsel,below,volu

/com Add a radius to the Low-Field Side of the wedge lip (trig very approximate)

*if,r_case,ne,0,then

csys,12

ksel,r,loc,x,ri12+dr_iwp-t,ri12+dr_iwp+t

*get,th12mx,kp,,mxloc,y

ksel,r,loc,y,th12mx-t,th12mx+t

csys

*get,x12mx,kp,,mxloc,x

*get,z12mx,kp,,mxloc,z

local,112,1,x12mx-1.05*(r_case+r_case*(1/sin(th12mx))),,z12mx-r_case

!local,112,1,x12mx-r_case*cos(th12mx)-(r_case+r_case*(sin(th12mx)))*tan(th12mx),,z12mx-r_case

wpcsys

wprot,,-90

wprot,th12mx

cylind,r_case,3*r_case,,dz_iwp/2,-90-th12mx,0

vsbv,case,all

*endif

/com

/com Reflect other wedges

/com

csys

VSYMM,Y,all

VSYMM,Z,all

/com

/com Treat WP/Wedge Interface

/com

allsel

vglue,all

/com

/com Slit Outboard Leg for Volt BC

/com

csys

wpcsys

allsel

cm,tf,volu

block,r0,3*r0,-dz_wp,dz_wp,,t_gap

btol,t_gap/2

vsbv,tf,all

/com

/com Cut Model in half for reflection later

/com

allsel

csys

wpcsys

cm,tfcoil1,volu

block,,10,-10,,-10,10

vsbv,tfcoil1,all

/com

/com Set Attributes

/com

*do,j,11,20

csys,j

ksel,s,loc,x,ri%j%-t,ri%j%+dr_iwp+t

*if,j,eq,11,then

ksel,r,loc,y,-dy%j%/2-t,dy%j%/2+t

*else

ksel,r,loc,y,-t,dy%j%+t

*endif

lslk,,1

asll,,1

vsla,,1

vatt,2,,3,j

*enddo

vsel,s,mat,,2

vsel,invert

vatt,16,,1

/com

/com Mesh the smeared WP Volumes

/com

MSHAPE,1,3D 

esize,cel

allsel

vmesh,all

/com

/com Reflect other half

/com

allsel

*get,dn1,node,,num,max

csys

nsym,y,dn1,all

esym, ,dn1,all

numm,node

/com

/com Make all TF Coils

/com

allsel

*get,dn2,node,,num,max

csys,1

ngen,ntf/nsym,dn2,all,,,,360/ntf

egen,ntf/nsym,dn2,all

*get,thmn,node,,mnloc,y

*get,thmx,node,,mxloc,y

modmesh,detach

ngen,2,,all,,,,-(thmn+thmx)/2

*get,thmn,node,,mnloc,y

*get,thmx,node,,mxloc,y

/com

/com Couple & Ground Volts and Apply Current to TF WP

/com

*do,j,1,ntf/nsym

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,x,r0,12

nsel,r,loc,z

nsel,r,loc,y,-(360/nsym/2)+(360/ntf)*(j-1),-(360/nsym/2)+(360/ntf)*(j-0)

cp,next,volt,all

*get,n_tf%j%,node,,num,min

f,n_tf%j%,amps,i_tf

/com Ground other end

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,x,r0,12

nsel,r,loc,z,t_gap

nsel,r,loc,y,-(360/nsym/2)+(360/ntf)*(j-1),-(360/nsym/2)+(360/ntf)*(j-0)

d,all,volt

*enddo

/com

/com Modular Coil Model

/com

/input,modcoils1,cdb     ! pull in PDR modular coils from HM Fan's file-1.7t.db

n_mcoils=3               ! number of Modular Coil Pairs

tpmc1=10                 ! turns per MC1 winding pack, 2 WP per coil, 4 in-hand

tpmc2=10                 ! turns per MC2 winding pack, 2 WP per coil, 4 in-hand

tpmc3=9                  ! turns per MC3 winding pack, 2 WP per coil, 4 in-hand

/com Node numbers for current application

n_m11=13599

n_m12=13609

n_m13=13619

n_m14=15619

n_m21=21639

n_m22=21649

n_m23=21659

n_m24=23659

n_m31=29679

n_m32=29689

n_m33=29699

n_m34=31699

/com Make nodal components out of Modular Coil Current Nodes

*do,jj,1,n_mcoils

nsel,none

*do,j,1,4

nsel,a,node,,n_m%jj%%j%

*enddo

cm,n_m%jj%,node

*enddo

/com

/com Modular Coil Currents

/com

cmsel,s,n_m1 $f,all,amps,i_mc1*tpmc1       ! MC1

cmsel,s,n_m2 $f,all,amps,i_mc2*tpmc2       ! MC2

cmsel,s,n_m3 $f,all,amps,i_mc3*tpmc3       ! MC3

/com Nix MAG DOF

esel,s,type,,2,3

nsle

d,all,mag

/com Set TEMP

esel,s,type,,1,3,2

nsle

d,all,temp,dtmp

allsel

bfunif,temp,dtmp

allsel

/psym,csys

nit=nint(i_tf/1000)

nim1=nint(i_mc1*tpmc1*2/1000)

nim2=nint(i_mc2*tpmc2*2/1000)

nim3=nint(i_mc3*tpmc3*2/1000)

/title,tfmodb3%rn%, NI(TF/MC1/MC2/MC3):%nit%/%nim1%/%nim2%/%nim3% kA-t

allsel

/edge,1,1

eplo

esel,s,mat,,2

esel,a,mat,,103,105

nsle

eplo

/edge

allsel

save

fini

/solu

tref,0

/com Solve for the conduction problem

allsel

esel,u,type,,1

*if,i_mc1+i_mc2+i_mc3,eq,0,then

esel,u,mat,,103,105  ! Nix MCs

*endif

nsle

solve

/com Solve for the fields

esel,s,mat,,2

nsle

*if,i_mc1+i_mc2+i_mc3,ne,0,then

esel,a,mat,,103,105

*endif

nsle

biot,new

allsel

solve

fini

:1000

/post1

/auto

/title,tfmodb3%rn%, NI(TF/MC1/MC2/MC3):%nit%/%nim1%/%nim2%/%nim3% kA-t

esel,s,mat,,2

esel,a,mat,,103,105

nsle

/view,1,1

/edge,1,1

eplo

/edge

/com Modular Coil Fields

esel,s,mat,,103,105

nsle

plns,b,sum

/com TF Coil Fields

esel,s,mat,,2

nsle

plns,b,sum

/com TF Coil Forces

*dim,f_t,,6,ntf/nsym

*do,j,1,ntf/nsym

th%j%=-(360/nsym/2)+arc/2+(360/ntf)*(j-1)

local,1000+j,,,,,th%j%

/psym,csys

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,y,th%j%-arc/2,th%j%+arc/2

esln,,1

rsys,1000+j

etab,fx,fmag,x

etab,fy,fmag,y

etab,fz,fmag,z

csys,1000+j

etab,xc,cent,x 

etab,yc,cent,y  

etab,zc,cent,z  

sadd,eccx,xc,,-1,,r0

!sadd,eccx,xc,,-1,,1.48736      ! Art's Centroid

/com Calculate Torques from each coil

/com Tx

smult,tx1,fy,zc,-1 

smult,tx2,fz,yc,+1

sadd,tx,tx1,tx2

/com Ty

smult,ty1,fx,zc,+1 

smult,ty2,fz,eccx,+1

sadd,ty,ty1,ty2

/com Tz

smult,tz1,fy,xc,+1 

smult,tz2,fx,yc,-1

sadd,tz,tz1,tz2

ssum

*get,fx,ssum,,item,fx

*get,fy,ssum,,item,fy

*get,fz,ssum,,item,fz

*get,tx,ssum,,item,tx

*get,ty,ssum,,item,ty

*get,tz,ssum,,item,tz

f_t(j,1)=fx

f_t(j,2)=fy

f_t(j,3)=fz

f_t(j,4)=tx

f_t(j,5)=ty

f_t(j,6)=tz

nfx=nint(fx)

nfy=nint(fy)

nfz=nint(fz)

kfx=0.01*nint(fx/10)

kfy=0.01*nint(fy/10)

kfz=0.01*nint(fz/10)

thj=th%j%

/title,tfmodb3%rn%, TF#%j% (%thj% deg), Local FX/FY/FZ=%kfx%/%kfy%/%kfz% kN

/title,tfmodb3%rn%, TF#%j% (%thj% deg), Local FX/FY/FZ=%nfx%/%nfy%/%nfz% N

/view,1,,1

plns,b,sum

*enddo

*vwrite,

('  F(radial)     F(OOP)    F(Vertical)  T(radial)    T(R0)   T(Vertical)')

*vwrite,f_t(1,1),f_t(1,2),f_t(1,3),f_t(1,4),f_t(1,5),f_t(1,6)

(1p6e12.4)

allsel

save

fini

!/exit,nosa

!/eof

:2000

srn=7

/filnam,tfmods3%srn%

/show,tfmods3%srn%,grp

/com 30: 4000 lb preload, Art's 08/03 currents, 2 load steps (just like previous, non-wedged, analysis)

/com 32: 

/com 33: 4000 lb preload, new smeared props, Not Converged

/com 34: 4000 lb preload, new smeared props

/com 35: 4000 lb preload, new smeared props

/com 36: 2000 lb preload, new smeared props

/com 37: 1000 lb preload, new smeared props

/com

/prep7

shpp,off

dtmp=0

k_nl=1                   ! 0: Wedged surfaces held to UY=0, 1: Generate Flex-Rigid Contact Interface

mu_sw=0.3                ! sidewall-to-sidewall friction coefficient

f_preload=1000/0.2248    ! applied radial preload per TF Bracket x 2 brackets per coil

/com Change to structural elements

et,1,92

et,3,92

/com Delete MC elements

esel,s,type,,2

nsle

edele,all

ndele,all

etdele,2

/com TF Coils (close to a simple toroidal system)

local,103,3,,,,,,,1.4

mp, kxx,2,1

mp,murx,2,1

mp,rsvx,2,1

mp,  ex,2,42.3e9

mp,  ey,2,41.2e9

mp,  ez,2,76.6e9

mp,alpx,2,9.5e-6

mp,alpy,2,9.6e-6

mp,alpz,2,13.0e-6

mp,nuxy,2,0.319

mp,nuyz,2,0.284

mp,nuxz,2,0.284

mp, gxy,2,2.1e9

mp, gyz,2,2.1e9   ! required to get beam bending right

mp, gxz,2,2.1e9   ! required to get beam bending right

esel,s,mat,,2

emodif,all,esys,103

csys

/com Apply Temps to TF Coil

esel,all

nsle

bfunif,temp,dtmp

/com Apply Wedged Face BCs

*if,k_nl,eq,0,then

/com Couple adjacent nodes and cyclically couple wedged surfaces at theta edges

csys,1

nsel,s,loc,x,,r0

cpintf,ux

cpintf,uy

cpintf,uz

allsel

nsel,u,loc,y,thmn+th,thmx-th

nrotate,all

/com Cyclic Coupling at inboard legs

cpcyc,ux,0.1*k,1,,thmx-thmn,,1

cpcyc,uy,0.1*k,1,,thmx-thmn,,1

cpcyc,uz,0.1*k,1,,thmx-thmn,,1

*else

/com Generate Node-Node Contact surfaces between Wedged Surfaces for nonlinear analyses

!et,11,178,,,1,1,0        ! contact elements

!et,11,178,,1,1,1,0        ! Lagrange Mult.

et,11,178,,2,1,1,0        ! 

mp,mu,11,mu_sw

mat,11

type,11

*do,j,1,ntf/nsym-1

/gopr

csys,1000+j

wpcsys

wprot,+arc/2

cswpla,2000+j

nsel,s,loc,y,-t,t

csys,1

*get,thmn%j%,node,,mnloc,y

*get,thmx%j%,node,,mxloc,y

th%j%=(thmn%j%+thmx%j%)/2

esel,s,mat,,3

!*if,j,ge,ntf/nsym/2-1,and,j,le,ntf/nsym/2+1,then

*if,j,ge,ntf/nsym/2,and,j,le,ntf/nsym/2,then

real,10+j

r,10+j,,0,2,,,-sin(th%j%)

rmore,cos(th%j%),,!2,2      ! 9 & 10 are scale factors on toln & ftol 

eint,,,low

*else

cpintf,ux

cpintf,uy

cpintf,uz

*endif

*enddo

/com Cyclic Coupling at inboard legs

csys,1

nsel,s,loc,x,,r0

allsel

nsel,u,loc,y,thmn+th,thmx-th

nrotate,all

/com Cyclic Coupling at inboard legs

cpcyc,ux,0.1*k,1,,thmx-thmn,,1

cpcyc,uy,0.1*k,1,,thmx-thmn,,1

cpcyc,uz,0.1*k,1,,thmx-thmn,,1

*endif

/com Restrain TF Coil in Z

*do,j,1,ntf/nsym

csys,1000+j

nsel,s,loc,y,-cel/2,cel/2

*if,j,le,ntf/nsym/2,then

*get,zgrnd,node,,mnloc,z

*else

!*get,zgrnd,node,,mxloc,z

*get,zgrnd,node,,mnloc,z

*endif

nsel,r,loc,z,zgrnd-t,zgrnd+t

*get,xmn_bot,node,,mnloc,x

*get,xmx_bot,node,,mxloc,x

xav_bot=(xmn_bot+xmx_bot)/2

nsel,r,loc,x,xav_bot-cel/2,xav_bot+cel/2

d,all,uz

*enddo

/com Apply Coupling at Equatorial Slit

csys,1

nsel,s,loc,x,r0,2*r0

nsel,r,loc,z,-t,2*t_gap

cpint,ux,2*t_gap

cpint,uy,2*t_gap

cpint,uz,2*t_gap

/com

/com Apply Radial Preload and lateral constraint

/com

*do,j,1,ntf/nsym

csys,1000+j

esel,all

nsel,ext

nsel,r,loc,y,-dz_iwp,dz_iwp

nsel,r,loc,x,r0,12

nsel,r,loc,z,z_preload-cel,z_preload+cel

*get,nnodes,node,,count

nrotate,all

f,all,fx,-f_preload/nnodes

nsle

nsel,ext

nsel,r,loc,y,-dz_iwp,dz_iwp

nsel,r,loc,x,r0,12

nsel,r,loc,z,-z_preload-cel,-z_preload+cel

*get,nnodes,node,,count

nrotate,all

f,all,fx,-f_preload/nnodes

/com Lateral Constraints

allsel

nsel,ext

nsel,r,loc,y,-dz_iwp/2-t,dz_iwp/2+t

nsel,r,loc,x,r0-0.1,r0+0.1

nsel,u,loc,y,-dz_iwp/2+t,dz_iwp/2-t

nrotate,all

d,all,uy

*enddo

allsel

save

fini

/solu

allsel

/com Solve the preload problem

*if,k_nl,eq,0,then

solve

*else

nsubst,10000,200000,4

autots,on

kbc,0

nropt,unsym     ! bombs on "Insufficient memory error during solution" with -m 128/64

!eqslv,pcg  ! PCG solver fails to solve problems with Lagrange multiplier method.

eqslv,sparse

!nropt,full,,off

!lnsrch,on

solve

*endif

fcum,add

ldread,forc,2,,,,tfmodb3%rn%,rst

nsubst,100,200000,4

solve

fini

:3000

/show,tfmods3%srn%,grp

/post1

set,last

/auto

/psym,csys

/cont,,,auto

/title,tfmods3%srn%, Stress Analysis

/view,1,,-1

pldi

esel,s,mat,,2

nsle

plns,s,int

rsys,1

plns,u,x

*if,k_nl,eq,1,then

/com Ratio of Shear Force to Notmal Force

esel,s,type,,11

*get,nce,elem,,count

etab,f_n,smisc,1

etab,f_y,smisc,2

etab,f_z,smisc,3

smult,f_y2,f_y,f_y

smult,f_z2,f_z,f_z

sadd,fy2fz2,f_y2,f_z2

sexp,f_s,fy2fz2,,0.5

sexp,muj,f_s,f_n,,-1

esel,r,etab,muj,0.99*mu_sw,1.01*mu_sw

*get,nsce,elem,,count

f_sl=0.001*nint(1000*nsce/nce)

esel,s,type,,11

esel,r,etab,f_n,0

*get,noce,elem,,count

f_op=0.001*nint(1000*noce/nce)

esel,s,type,,11

esel,r,etab,muj,0.01*mu_sw,0.99*mu_sw

*get,nstkce,elem,,count

f_stk=0.001*nint(1000*nstkce/nce)

esel,s,type,,11

esel,r,etab,muj,0.01*mu_sw,1.01*mu_sw

*get,ncntce,elem,,count

ssum

*get,mursum,ssum,,item,muj

muav=0.01*nint(100*mursum/ncntce)

/title,tfmods3%srn%, Stuck: %f_stk%, Sliding: %f_sl%, Open: %f_op%, Mu(ave)=%muav%

esel,s,type,,11

nsle

plet,muj

/com Max Slip

esel,s,type,,11

etab,uty,nmisc,6

etab,utz,nmisc,7

smult,uty2,uty,uty

smult,utz2,utz,utz

sadd,uty2utz2,uty2,utz2

sexp,ut,uty2utz2,,.5

plet,ut

*endif

allsel

/cont

fini

/exit,all

/eof

